

CASE STUDY

twenty
**Stochastic Customer
Forecasting**

case study
OVERVIEW

CS20.1	Application Overview and Model Development
CS20.2	Worksheets
CS20.3	User Interface
CS20.4	Procedures
CS20.5	Re-solve Options
CS20.6	Summary
CS20.7	Extensions

CS20.1 *Application Overview and Model Development*

This application is to forecast stochastic customer arrival while considering a set of factors that influence the forecast. We are looking at a bank scenario in which we are trying to forecast the number of daily customers that arrive at the bank. We consider the following factors that affect this forecast: the month, the day of the week, and such special factors as holidays, periodic days (such as paydays), and ranges of dates (such as Christmas week). Since this stochastic forecasting, also known as regression forecasting, requires up to 15 independent variables, we use a non linear programming model and the Solver as alternatives in order to estimate the coefficients of these factors.

CS20.1.1 **Model Definition and Assumptions**

This application determines a constant factor, a factor for each month, a factor for each weekday, and special factors such that the sum of the square errors between the predicted number of customers and the historical number of customers is minimized. To forecast the number of customers with these factors, we use the following equation:

Predicted Number of Customers = Constant Factor + Month Factor +
Weekday Factor + Sum(Special Factors)

With the decision variable cells, the month number, and the weekday value for each calendar day, we use LOOKUP functions to insert the corresponding month and weekday factors into this formula. We also use the SUMPRODUCT function to insert the corresponding special factors with a binary representation of which special factors are applied to a particular date. Below is an example of how this formula is written in Excel. (We consider month 1 = January and weekday 2 = Tuesday.)

=CFactor + HLOOKUP(1, MonthLook, 2)+HLOOKUP(2, WeekLook, 2)+ SUMPRODUCT
(SpecialFactors, \$T\$22:\$Z\$22)

To insert this formula for every calendar day given in the user's historical data, we loop over the data and dynamically enter the formula using the month and weekday in the corresponding row of data. We also enter the formula for the sum of squared errors between this forecasted number of customers and the historical number of customers. In the "Procedures" section, we will review these formulas, and in the "Worksheets" section, we will explain the layout of the model sheet.

```
Do While Range("DataStart").Offset(i, 0).Value <> ""
  With Range("DataStart").Offset(i, 5).Formula = "=CFactor + HLOOKUP(" &
.Offset(i, 1).Value & ",MonthLook,2) + " & "HLOOKUP(" & .Offset(i, 2).Value &
",WeekLook,2)+SUMPRODUCT(SpecialFactors," & Range("SPValues").Offset(i,
0).Address & ")".Offset(i, 6).FormulaR1C1 = "=(RC[-2] - RC[-1])^2"
  End With
  i = i + 1
Loop
```

Please see Microsoft Excel: Data Analysis and Business Modeling by Winston for more details.

CS20.1.2 Input

The input for this application is the following:

- Historical data = the number of customers on each workday for an entire calendar year
- Special factors = the holidays, periodic days, or a range of dates with a biased number of customers

CS20.1.3 Output

The output for this application is the following:

- A constant factor, month factors, weekday factors, and special factors
- A predicted number of customers
- The square error between the predicted number of customers and the historical data
- A chart of predicted and historical values

CS20.2 Worksheets

This application requires four sheets: the welcome sheet, the historical data sheet, the solutions sheet, and the solutions chart sheet. The welcome sheet contains the title, the description of the application, and the “Run Demo” and “Start” buttons. (See Figure CS20.1.) The “Run Demo” button copies the demo data to the historical data sheet and takes the user to this sheet. The “Start” button prompts the user for some input and then takes him or her to the historical data sheet.




Figure CS20.1 The welcome sheet.

The historical data sheet contains the historical number of customers for every workday for an entire calendar year. (See Figure CS20.2.) The user can change the calendar year using the spin button at the top of the sheet; the calendar day, month, and weekday

values are automatically updated for the selected year. The user can also import this information as a text file. Once the historical data has been entered, the user presses the “Continue” button to proceed to the solution sheet.

Historical Data			
Calendar Year 2001			
Calendar Day	Month (Jan = 1, Dec = 12)	Week Day (Mon = 1, Fri = 5)	Number of Customers
1/2/2001	1	2	1,825
1/3/2001	1	3	1,257
1/4/2001	1	4	969
1/5/2001	1	5	1,672
1/8/2001	1	1	1,098
1/9/2001	1	2	691
1/10/2001	1	3	672
1/11/2001	1	4	754
1/12/2001	1	5	972
1/15/2001	1	1	816
1/16/2001	1	2	717
1/17/2001	1	3	728
1/18/2001	1	4	711
1/19/2001	1	5	1,545
1/22/2001	1	1	873
1/23/2001	1	2	713
1/24/2001	1	3	626
1/25/2001	1	4	653
1/26/2001	1	5	1,080
1/29/2001	1	1	650

Figure CS20.2 The historical data sheet.

The solution sheet contains the model and the results of the model for the factors in the customer forecasting. (See Figure CS20.3.) A constant factor, a factor for each month, a factor for each weekday, and special factors are determined such that the sum of the mean square errors between the predicted number of customers and the historical number of customers is minimized. Outliers in the mean square error column are highlighted to help the user determine if another special factor should be considered. The user can press the “View Chart” button to view the solution chart sheet.

The Solution																																																																																																																																																	
Constant Factor	C 801.74																																																																																																																																																
Month Factors	<table border="1"> <tr><th></th><th>1</th><th>2</th><th>3</th><th>4</th><th>5</th><th>6</th><th>7</th><th>8</th><th>9</th><th>10</th><th>11</th><th>12</th><th>Average</th></tr> <tr><td></td><td>-14.79</td><td>-7.75</td><td>52.80</td><td>48.13</td><td>174.65</td><td>152.22</td><td>166.46</td><td>112.24</td><td>-25.82</td><td>12.40</td><td>37.00</td><td>94.23</td><td>66.81</td></tr> </table>		1	2	3	4	5	6	7	8	9	10	11	12	Average		-14.79	-7.75	52.80	48.13	174.65	152.22	166.46	112.24	-25.82	12.40	37.00	94.23	66.81																																																																																																																				
	1	2	3	4	5	6	7	8	9	10	11	12	Average																																																																																																																																				
	-14.79	-7.75	52.80	48.13	174.65	152.22	166.46	112.24	-25.82	12.40	37.00	94.23	66.81																																																																																																																																				
Week Day Factors	<table border="1"> <tr><th></th><th>1</th><th>2</th><th>3</th><th>4</th><th>5</th><th>Average</th></tr> <tr><td></td><td>226.67</td><td>-9.28</td><td>-41.64</td><td>14.81</td><td>611.29</td><td>160.35</td></tr> </table>		1	2	3	4	5	Average		226.67	-9.28	-41.64	14.81	611.29	160.35																																																																																																																																		
	1	2	3	4	5	Average																																																																																																																																											
	226.67	-9.28	-41.64	14.81	611.29	160.35																																																																																																																																											
Special Factors	<table border="1"> <tr><th></th><th>New Year's Eve</th><th>Christmas Eve</th><th></th><th>Payday</th><th></th><th>Christmas Week</th><th></th></tr> <tr><td></td><td>1</td><td>2</td><td>3</td><td>4</td><td>5</td><td>6</td><td>7</td></tr> <tr><td></td><td>0.00</td><td>0.00</td><td>0.00</td><td>0.00</td><td>0.00</td><td>0.00</td><td>0.00</td></tr> </table>		New Year's Eve	Christmas Eve		Payday		Christmas Week			1	2	3	4	5	6	7		0.00	0.00	0.00	0.00	0.00	0.00	0.00																																																																																																																								
	New Year's Eve	Christmas Eve		Payday		Christmas Week																																																																																																																																											
	1	2	3	4	5	6	7																																																																																																																																										
	0.00	0.00	0.00	0.00	0.00	0.00	0.00																																																																																																																																										
Calendar Year	2001																																																																																																																																																
<table border="1"> <thead> <tr> <th>Calendar Day</th> <th>Month (Jan = 1, Dec = 12)</th> <th>Week Day (Mon = 1, Fri = 5)</th> <th>Special Factors</th> <th>Historical Number of Customers</th> <th>Predicted Number of Customers</th> <th>Mean Square Error</th> <th>SUM of Mean Square Errors</th> </tr> </thead> <tbody> <tr><td>1/2/2001</td><td>1</td><td>2</td><td></td><td>1,825</td><td>777.67</td><td>1,096.897.46</td><td>12,917,825.65</td></tr> <tr><td>1/3/2001</td><td>1</td><td>3</td><td></td><td>1,257</td><td>745.31</td><td>261,628.77</td><td></td></tr> <tr><td>1/4/2001</td><td>1</td><td>4</td><td></td><td>969</td><td>801.76</td><td>27,969.05</td><td></td></tr> <tr><td>1/5/2001</td><td>1</td><td>5</td><td></td><td>1,672</td><td>1,398.24</td><td>74,943.19</td><td></td></tr> <tr><td>1/8/2001</td><td>1</td><td>1</td><td></td><td>1,098</td><td>1,013.52</td><td>7,136.39</td><td></td></tr> <tr><td>1/9/2001</td><td>1</td><td>2</td><td></td><td>691</td><td>777.67</td><td>7,511.91</td><td></td></tr> <tr><td>1/10/2001</td><td>1</td><td>3</td><td></td><td>672</td><td>745.31</td><td>5,374.05</td><td></td></tr> <tr><td>1/11/2001</td><td>1</td><td>4</td><td></td><td>754</td><td>801.76</td><td>2,281.07</td><td></td></tr> <tr><td>1/12/2001</td><td>1</td><td>5</td><td></td><td>972</td><td>1,398.24</td><td>181,682.65</td><td></td></tr> <tr><td>1/15/2001</td><td>1</td><td>1</td><td></td><td>816</td><td>1,013.52</td><td>39,015.28</td><td></td></tr> <tr><td>1/16/2001</td><td>1</td><td>2</td><td></td><td>717</td><td>777.67</td><td>3,681.00</td><td></td></tr> <tr><td>1/17/2001</td><td>1</td><td>3</td><td></td><td>728</td><td>745.31</td><td>299.56</td><td></td></tr> <tr><td>1/18/2001</td><td>1</td><td>4</td><td></td><td>711</td><td>801.76</td><td>8,237.47</td><td></td></tr> <tr><td>1/19/2001</td><td>1</td><td>5</td><td></td><td>1,545</td><td>1,398.24</td><td>21,537.77</td><td></td></tr> <tr><td>1/22/2001</td><td>1</td><td>1</td><td></td><td>873</td><td>1,013.52</td><td>19,746.67</td><td></td></tr> <tr><td>1/23/2001</td><td>1</td><td>2</td><td></td><td>713</td><td>777.67</td><td>4,182.37</td><td></td></tr> <tr><td>1/24/2001</td><td>1</td><td>3</td><td></td><td>626</td><td>745.31</td><td>14,234.38</td><td></td></tr> </tbody> </table>	Calendar Day	Month (Jan = 1, Dec = 12)	Week Day (Mon = 1, Fri = 5)	Special Factors	Historical Number of Customers	Predicted Number of Customers	Mean Square Error	SUM of Mean Square Errors	1/2/2001	1	2		1,825	777.67	1,096.897.46	12,917,825.65	1/3/2001	1	3		1,257	745.31	261,628.77		1/4/2001	1	4		969	801.76	27,969.05		1/5/2001	1	5		1,672	1,398.24	74,943.19		1/8/2001	1	1		1,098	1,013.52	7,136.39		1/9/2001	1	2		691	777.67	7,511.91		1/10/2001	1	3		672	745.31	5,374.05		1/11/2001	1	4		754	801.76	2,281.07		1/12/2001	1	5		972	1,398.24	181,682.65		1/15/2001	1	1		816	1,013.52	39,015.28		1/16/2001	1	2		717	777.67	3,681.00		1/17/2001	1	3		728	745.31	299.56		1/18/2001	1	4		711	801.76	8,237.47		1/19/2001	1	5		1,545	1,398.24	21,537.77		1/22/2001	1	1		873	1,013.52	19,746.67		1/23/2001	1	2		713	777.67	4,182.37		1/24/2001	1	3		626	745.31	14,234.38		View Chart
Calendar Day	Month (Jan = 1, Dec = 12)	Week Day (Mon = 1, Fri = 5)	Special Factors	Historical Number of Customers	Predicted Number of Customers	Mean Square Error	SUM of Mean Square Errors																																																																																																																																										
1/2/2001	1	2		1,825	777.67	1,096.897.46	12,917,825.65																																																																																																																																										
1/3/2001	1	3		1,257	745.31	261,628.77																																																																																																																																											
1/4/2001	1	4		969	801.76	27,969.05																																																																																																																																											
1/5/2001	1	5		1,672	1,398.24	74,943.19																																																																																																																																											
1/8/2001	1	1		1,098	1,013.52	7,136.39																																																																																																																																											
1/9/2001	1	2		691	777.67	7,511.91																																																																																																																																											
1/10/2001	1	3		672	745.31	5,374.05																																																																																																																																											
1/11/2001	1	4		754	801.76	2,281.07																																																																																																																																											
1/12/2001	1	5		972	1,398.24	181,682.65																																																																																																																																											
1/15/2001	1	1		816	1,013.52	39,015.28																																																																																																																																											
1/16/2001	1	2		717	777.67	3,681.00																																																																																																																																											
1/17/2001	1	3		728	745.31	299.56																																																																																																																																											
1/18/2001	1	4		711	801.76	8,237.47																																																																																																																																											
1/19/2001	1	5		1,545	1,398.24	21,537.77																																																																																																																																											
1/22/2001	1	1		873	1,013.52	19,746.67																																																																																																																																											
1/23/2001	1	2		713	777.67	4,182.37																																																																																																																																											
1/24/2001	1	3		626	745.31	14,234.38																																																																																																																																											

Figure CS20.3 The solution sheet.

The solution chart sheet graphs the predicted number of customers with the historical number of customers. (See Figure CS20.4.) The user presses the “Return to Results” button to return to the solutions sheet.





Figure CS20.4 The solution chart sheet.

 Summary	Welcome sheet	Contains the application description and the “Run Demo” and “Start” buttons.
	Historical data sheet	Where the user enters the historical number of customers for every workday of an entire calendar year.
	Solution sheet	Contains the factors and predicted values resulting from the optimization model.
	Solution chart sheet	Contains the chart of predicted and historical number of customers.

CS20.3 *User Interface*

For this application’s user interface, we use navigational buttons, controls on the worksheet, and two user forms. By pressing the “Start” button on the welcome sheet, the user is shown a historical data user form. (See Figure CS20.5.) This form, which requires two option buttons and a frame, determines if the user will import data from a text file or enter the historical number of customers manually.



Historical Data

Please select one of the following methods to enter your historical data.

Import Text File


Enter Manually

Note: Historical data should contain the following: Calendar Day (1-365), Month (1-12), Week Day (1-5), and Number of Customers.

OK Cancel

Figure CS20.5 The historical data form.

The user then views the historical data sheet where there is a spin button control that allows the user to change the calendar year to study. (See Figure CS20.6.) When this spin button is changed, the linked cell is updated, and the calendar days, the month, and the weekday values are automatically updated.



Historical Data Exit Continue

Calendar Year 2004 ◀ ▶

Calendar Day	Month (Jan = 1, Dec = 12)	Week Day (Mon = 1, Fri = 5)	Number of Customers
1/3/2004	1	1	
1/4/2004	1	2	
1/5/2004	1	3	
1/6/2004	1	4	
1/7/2004	1	5	
1/10/2004	1	1	

Figure CS20.6 The historical data year spin button.

When the user presses the “Continue” button on this sheet, he or she views the special factors user form. (See Figure CS20.7.) This form prompts the user to select any special factors that may bias his or her historical number of customers. There are three categories of special factors: holidays, periodic days (per month), and ranges of dates. For each category, some default days are provided as well as a blank day in which the user can create his or her own special factors. The user can select all or none of these and then click the “Solve” button to solve the model and proceed to the solution sheet. Frames group these categories of special factors, and check boxes are provided for each special factor selection. Additionally, combo boxes are provided for the months, days, and periodic type and day.

Special Factors

Please select all of the special factors which you want to consider. You may add some new special factors by specifying the dates that apply.

Holidays

New Years Eve December 31

Christmas Eve December 24

Other1 [Month] [Day]

Periodic Days (per month)

Payday Last [Day]

Other2 [Month] [Day]

Ranges

Christmas Week From
December 25
To
January 1

Other3 From
[Month] [Day]
To
[Month] [Day]

Solve Cancel

Figure CS20.7 The special factors form.

 <p>Summary</p>	Historical data form	Determines how the user enters the historical data.
	Year spin button	Allows the user to change the year of the historical data and to automatically update the calendar days.
	Special factors form	The user can select any or all of the special factors that may bias his or her historical data: holidays, periodic days, or ranges of dates.
	Navigational buttons	“End,” “View Chart,” and “Return to Results” buttons.
	Functional buttons	“Run Demo,” “Start,” “Continue,” and “Re-solve.”

CS20.4 *Procedures*

We will now outline the procedures for this application beginning with the variable declarations and the initial sub procedures. (See Figure CS20.8.) The “Main” procedure, which is called from the “Start” button, calls the “ClearPrev” procedure to clear the previous values from all the sheets. It then initializes the historical data calendar dates on the historical data sheet and takes the user to this sheet. The historical data form then appears. The “Demo” procedure, which is called from the “Run Demo” button, copies the demo year and data to the historical data sheet and takes the user to this sheet.

```

Option Explicit
Option Base 1

Public i As Integer, j As Integer, _
HolSF(3, 4) As Variant, PerSF(2, 4) As Variant, RangSF(2, 4, 2) As Variant, _
StdDevMSE As Double, MeanMSE As Double

Sub Main() 'called from the Start button
    Call ClearPrev

    'initialize historical data calendar dates
    Worksheets("HistData").spnYear.Value = 2004
    j = 0
    i = 1
    Do While j < 360
        If Weekday(DateAdd("d", j, 1 / 1 / 2004), vbMonday) < 6 Then
            With Range("HistDataStart")
                .Offset(i, 0).Value = DateAdd("d", j, "1 / 1 / 2004")
                .Offset(i, 1).Value = Month(DateAdd("d", j, 1 / 1 / 2004))
                .Offset(i, 2).Value = Weekday(DateAdd("d", j, 1 / 1 / 2004), vbMonday)
            End With
            i = i + 1
        End If
        j = j + 1
    Loop

    Worksheets("HistData").Visible = True
    Worksheets("Welcome").Visible = False
    Range("A1").Select

    frmHistData.Show
End Sub

Sub Demo() 'called from Run Demo button
    Call ClearPrev
    'copy demo year
    Worksheets("ExHistData").Range("ExYear").Copy
    Worksheets("HistData").Range("HistYear").PasteSpecial
    'copy demo data
    Worksheets("ExHistData").Range("ExData").Copy
    Worksheets("HistData").Range("HistDataStart").Offset(1, 0).PasteSpecial
    Application.CutCopyMode = False

    Worksheets("HistData").Visible = True
    Worksheets("Welcome").Visible = False
    Range("A1").Select
End Sub

```

Figure CS20.8 The variable declarations and the “Main” and “Demo” procedures.

The “ClearPrev” procedure clears the previous values and initializes the special factors arrays. (See Figure CS20.9.) There are three special factors arrays: one for the holidays; one for the periodic days; and one for the ranges of dates. The default dates are set, and the Boolean values for the user selection are reset.

```

Sub ClearPrev() 'clear previous values and initialize special factors arrays
Worksheets("HistData").Range("HistData").ClearContents
With Worksheets("Model")
    Range(Range("DataStart").Offset(1, 0), Range("DataStart").Offset(1, 6).End(xlDown)).ClearContents
    Range("SPValues").Columns.Hidden = False
    Range(Range("SFStart").Offset(1, 0), Range("SFStart").Offset(1, 6).End(xlDown)).ClearContents
    Range("SPFacStart").Offset(-2, 2).ClearContents
    Range("SPFacStart").Offset(-2, 4).ClearContents
    Range("SPFacStart").Offset(-2, 6).ClearContents
    Range("CFactor").ClearContents
    Range("MonthFactors").ClearContents
    Range("WeekFactors").ClearContents
    Range("SpecialFactors").ClearContents
    Range("MSE").Interior.ColorIndex = 2
End With

'holiday special factors dates
HolSF(1, 2) = 12 'New Years Eve
HolSF(1, 3) = 31
HolSF(1, 4) = 1
HolSF(2, 2) = 12 'Christmas Eve
HolSF(2, 3) = 24
HolSF(2, 4) = 2
HolSF(3, 4) = 3
'user selected
For i = 1 To 3
    HolSF(i, 1) = False
Next i

'periodic special factors
PerSF(1, 1) = False
PerSF(1, 4) = 4
PerSF(2, 1) = False
PerSF(2, 4) = 5

'range special factors
RangSF(1, 1, 1) = False
RangSF(2, 1, 1) = False
RangSF(1, 2, 1) = 12 'week start = Christmas Day
RangSF(1, 2, 2) = 25
RangSF(1, 3, 1) = 1 'week end = New Years Day
RangSF(1, 3, 2) = 1
RangSF(1, 4, 1) = 6
RangSF(2, 4, 1) = 7
End Sub

```

Figure CS20.9 The “ClearPrev” procedure.

The procedures for the historical data form prompt the user to select a text file if he or she chooses to import the historical data. Or, they take the user directly to the historical data sheet if he or she chooses to enter the data manually. (See Figure CS20.10.) The event procedure for the spin button on the historical data sheet updates the calendar dates, month values, and weekday values for the selected year. (See Figure CS20.11.)

```

Private Sub cmdCancel_Click()
    Unload Me
End Sub

Private Sub cmdOK_Click()
    If optImport Then 'import text file
        Unload Me

        Dim pathname As String
        pathname = Application.GetOpenFilename("TEXT (*.txt), *.txt", , "Historical Data")

        Worksheets("HistData").Visible = True
        Worksheets("Welcome").Visible = False
        Worksheets("HistData").Activate

        With ActiveSheet.QueryTables.Add(Connection:= _
            "TEXT;" & pathname & "", Destination:=Range("HistDataStart").Offset(1, 0))

            .PreserveFormatting = True
            .RefreshStyle = xlOverwriteCells

            .TextFileStartRow = 1
            .TextFileParseType = xlDelimited
            .TextFileCommaDelimiter = True

            .Refresh BackgroundQuery:=False
        End With

        ElseIf optManual Then 'take user to historical data sheet
            Unload Me
            Worksheets("HistData").Visible = True
            Worksheets("Welcome").Visible = False

            MsgBox "Please enter your data into the appropriate columns."
        End If
    End Sub

```

Figure CS20.10 The historical data form procedures.

```

Private Sub spnYear_Change() 'recalculates the calendar dates for a new year value
    Dim HistYear As Double, HistDate As Date
    HistYear = spnYear.Value
    HistDate = DateValue("1/1/" & HistYear & "")
    j = 0
    i = 1
    Do While j < 360
        If Weekday(DateAdd("d", j, HistDate), vbMonday) < 6 Then
            With Range("HistDataStart")
                .Offset(i, 0).Value = DateAdd("d", j, HistDate)
                .Offset(i, 1).Value = Month(DateAdd("d", j, HistDate))
                .Offset(i, 2).Value = Weekday(DateAdd("d", j, HistDate), vbMonday)
            End With
            i = i + 1
        End If
        j = j + 1
    Loop
End Sub

```

Figure CS20.11 The historical data sheet control event procedures.

The "Continue" procedure is called from the "Continue" button on the historical data sheet. (See Figure CS20.12.) This procedure copies the historical data and pastes it on the solution sheet for the model calculations. It then displays the special factors form and takes the user to the solution sheet. It ends by calling the "RecordSF" and "RunSolver" procedures.

```

Sub Continue() 'called from Continue button on historical data sheet
'check that values exist
If Range("HistDataStart").Offset(1, 3).Value = "" Then
MsgBox "You have not entered any data"
Exit Sub
End If
Range(Range(Range("HistDataStart").Offset(1, 3), _
Range("HistDataStart").Offset(1, 3).End(xlDown)), Range(Range("HistDataStart").Offset(1, 3), _
Range("HistDataStart").Offset(1, 3).End(xlDown)).End(xlToLeft)).Name = "HistData"
'copy hist data to model sheet
Worksheets("HistData").Range("HistData").Copy
Range("DataStart").Offset(1, 0).PasteSpecial
Range("HistYear").Copy
Range("CalYear").PasteSpecial
Range(Range("DataStart").Offset(1, 3), Range("DataStart").Offset(1, 3).End(xlDown)).Cut
ActiveSheet.Paste Destination:=Range("DataStart").Offset(1, 4)
Application.CutCopyMode = False

frmSpecFact.Show

Application.ScreenUpdating = False
Worksheets("Model").Visible = True
Worksheets("HistData").Visible = False

Call RecordSF
Call RunSolver

Range("A1").Select
Application.ScreenUpdating = True
End Sub

```

Figure CS20.12 The “Continue” procedure.

The procedures for the special factors form record all of the special factors selected by the user. (See Figures CS8.13 and CS8.14.) The default days for each category are checked as well as the user-defined special factors in each category. The application also updates the corresponding special factors arrays, and an initialization procedure sets the row sources of the form combo boxes and the default values.

The “RecordSF” procedure scans over the dates in the calendar from the historical data and marks which special factor is present. (See Figure CS20.15.) A mark is also made in the hidden special factors table, which is used in the model. For all possible days in each category of the special factors, the corresponding special factors array values are checked.

The “RunSolver” procedure solves the model. (See Figure CS20.16.) It inserts the formulas for the column to predict the number of customers and to calculate the squared error between the predicted and the historical values. “RunSolver” then names some ranges and inserts more formulas for the sum of the squared errors, the average month factors, and the average week factors. The Solver model is now prepared and run. The user can then observe the solutions to highlight outliers using the mean and standard deviation of the sum of the squared errors. Lastly, the application updates the data for the graph of the predicted and historical values.

```

Private Sub cmdCancel_Click()
    Unload Me
    Worksheets("Welcome").Visible = True
    Worksheets("HistData").Visible = False
End Sub
End Sub

Private Sub cmdOK_Click() 'record special factors
    'check holidays
    If chkHolOther Then
        'record month and day
        HolSF(3, 1) = True
        HolSF(3, 2) = cmbMonthHol.Value
        HolSF(3, 3) = CInt(cmbDayHol.Value)
        Range("SPFacStart").Offset(-2, HolSF(3, 4) - 1).Value = txtHolOther.Value
    End If

    'check payday
    If chkPayday Then
        PerSF(1, 1) = True
        PerSF(1, 2) = cmbFLPayday.Value
        PerSF(1, 3) = cmbDayPayday.Value
    End If

    If chkPeriodOther Then
        'record
        PerSF(2, 1) = True
        PerSF(2, 2) = cmbFLOther.Value
        PerSF(2, 3) = cmbDayOther.Value
        Range("SPFacStart").Offset(-2, PerSF(2, 4) - 1).Value = txtPeriodOther.Value
    End If

    'check ranges
    If chkRangeOther Then
        'record From and To dates
        RangSF(2, 1, 1) = True
        RangSF(2, 2, 1) = cmbMonFrom.Value
        RangSF(2, 2, 2) = CInt(cmbDayFrom.Value)
        RangSF(2, 3, 1) = cmbMonTo.Value
        RangSF(2, 3, 2) = CInt(cmbDayTo.Value)
        Range("SPFacStart").Offset(-2, RangSF(2, 4, 1) - 1).Value = txtRangeOther.Value
    End If

    Unload Me
End Sub

```

Figure CS20.13 The first part of the special factors form procedures.

```

Private Sub chkNewYr_Click()
    If chkNewYr Then
        HolSF(1, 1) = True
    Else
        HolSF(1, 1) = False
    End If
End Sub

Private Sub chkChristmas_Click()
    If chkChristmas Then
        HolSF(2, 1) = True
    Else
        HolSF(2, 1) = False
    End If
End Sub

Private Sub chkChrsWeek_Click()
    If chkChrsWeek Then
        RangSF(1, 1, 1) = True
    Else
        RangSF(1, 1, 1) = False
    End If
End Sub

'initialization

Private Sub UserForm_Initialize()
    cmbMonthHol.RowSource = "Months"
    cmbDayHol.RowSource = "Dates"

    cmbFLPayday.RowSource = "Periodic"
    cmbFLPayday.Value = cmbFLPayday.List(1)
    cmbDayPayday.RowSource = "Days"

    cmbFLOther.RowSource = "Periodic"
    cmbDayOther.RowSource = "Days"
    cmbMonFrom.RowSource = "Months"
    cmbDayFrom.RowSource = "Dates"
    cmbMonTo.RowSource = "Months"
    cmbDayTo.RowSource = "Dates"
End Sub

```

Figure CS20.14 The second part of the special factors form procedures.

```

Sub RecordSF() 'scan over dates to record special factors
j = 1
Do While Range("DataStart").Offset(j, 0).Value <> ""

    For i = 1 To 3
        If HolSF(i, 1) = True Then
            'check month and date
            If Range("DataStart").Offset(j, 1).Value = HolSF(i, 2) Then
                If Range("DataStart").Offset(j, 0).Value = HolSF(i, 3) Then
                    Range("DataStart").Offset(j, 3).Value = _
                        Range("DataStart").Offset(j, 3).Value & " " & HolSF(i, 4)
                    Range("SFStart").Offset(j, HolSF(i, 4) - 1).Value = 1
                End If
            End If
        End If
    Next i

    For i = 1 To 2
        If PerSF(i, 1) = True Then
            'check periodic
            If Range("DataStart").Offset(j, 2).Value = PerSF(i, 3) Then
                If PerSF(i, 2) = "First" Then
                    If Range("DataStart").Offset(j, 0).Value <= 7 Then
                        Range("DataStart").Offset(j, 3).Value = _
                            Range("DataStart").Offset(j, 3).Value & " " & PerSF(i, 4)
                        Range("SFStart").Offset(j, PerSF(i, 4) - 1).Value = 1
                    End If
                ElseIf PerSF(i, 2) = "Last" Then
                    If Range("DataStart").Offset(j, 0).Value >= 23 Then
                        Range("DataStart").Offset(j, 3).Value = _
                            Range("DataStart").Offset(j, 3).Value & " " & PerSF(i, 4)
                        Range("SFStart").Offset(j, PerSF(i, 4) - 1).Value = 1
                    End If
                End If
            End If
        End If
    Next i

    For i = 1 To 2
        If RangSF(i, 1, 1) = True Then
            'check range start to finish
            If Range("DataStart").Offset(j, 1).Value = RangSF(i, 2, 1) Then
                If (RangSF(i, 2, 1) = RangSF(i, 3, 1) And _
                    Range("DataStart").Offset(j, 0).Value <= RangSF(i, 3, 2)) Or _
                    (RangSF(i, 2, 1) <> RangSF(i, 3, 1) And _
                    Range("DataStart").Offset(j, 0).Value >= RangSF(i, 2, 2)) Then
                    Range("DataStart").Offset(j, 3).Value = _
                        Range("DataStart").Offset(j, 3).Value & " " & RangSF(i, 4, 1)
                    Range("SFStart").Offset(j, RangSF(i, 4, 1) - 1).Value = 1
                End If
            ElseIf Range("DataStart").Offset(j, 1).Value = RangSF(i, 3, 1) Then
                If Range("DataStart").Offset(j, 0).Value <= RangSF(i, 3, 2) Then
                    Range("DataStart").Offset(j, 3).Value = _
                        Range("DataStart").Offset(j, 3).Value & " " & RangSF(i, 4, 1)
                    Range("SFStart").Offset(j, RangSF(i, 4, 1) - 1).Value = 1
                End If
            End If
        End If
    Next i

    j = j + 1
Loop
End Sub

```

Figure CS20.15 The "RecordSF" procedure.

```

Sub RunSolver() 'solve the model
'insert formulas for predicted number of customers and squared errors
i = 1
Do While Range("DataStart").Offset(i, 0).Value <> ""
    With Range("DataStart")
        .Offset(i, 5).Formula = "=CFactor + HLOOKUP(" & .Offset(i, 1).Value & ",MonthLook,2) + " & _
            "HLOOKUP(" & .Offset(i, 2).Value & ",WeekLook, 2) + SUMPRODUCT(SpecialFactors, " & _
            Range("SPValues").Offset(i, 0).Address & ")"

        .Offset(i, 6).FormulaR1C1 = "=RC[-2] - RC[-1]^2"
    End With
    i = i + 1
Loop

'name ranges and insert more formulas
Range(Range("DataStart").Offset(1, 6), Range("DataStart").Offset(1, 6).End(xlDown)).Name = "MSE"
Range("SumMSE").Formula = "=SUM(MSE)"
Range("MonthAvg").Formula = "=AVERAGE(MonthFactors)"
Range("WeekAvg").Formula = "=AVERAGE(WeekFactors)"

'run Solver
SolverReset
SolverOK SetCell:=Range("SumMSE"), MaxMinVal:=2, _
    ByChange:=Range("CFactor, MonthFactors, WeekFactors, SpecialFactors")
SolverAdd CellRef:=Range("MonthAvg"), Relation:=2, FormulaText:=0
SolverAdd CellRef:=Range("WeekAvg"), Relation:=2, FormulaText:=0
SolverSolve UserFinish:=True

'highlight outliers
MeanMSE = Application.WorksheetFunction.Average(Range("MSE"))
StdDevMSE = Application.WorksheetFunction.StDev(Range("MSE"))
i = 1
Do While Range("DataStart").Offset(i, 0).Value <> ""
    If Range("DataStart").Offset(i, 6).Value >= (MeanMSE + StdDevMSE) _
        Or Range("DataStart").Offset(i, 6).Value <= (MeanMSE - StdDevMSE) Then
        Range("DataStart").Offset(i, 6).Interior.ColorIndex = 40
    End If
    i = i + 1
Loop

'update graph data
Application.Union(Range(Range("DataStart").Offset(1, 0), _
    Range("DataStart").Offset(1, 0).End(xlDown)), _
    Range("MSE").Offset(0, -1), Range("MSE").Offset(0, -2)).Name = "GraphData"
Sheets("ResultsChart").Activate
ActiveChart.SetSourceData Source:=Range("GraphData")
ActiveChart.SeriesCollection(1).Name = "Historical Values"
ActiveChart.SeriesCollection(2).Name = "Predicted Values"

Range("SPValues").Columns.Hidden = True
Worksheets("Model").Activate
Range("A1").Select
End Sub

```

Figure CS20.16 The “RunSolver” procedure.

Figure CS20.17 presents the navigational procedures.

```

Sub ExitProg()
    Worksheets("Welcome").Visible = True
    ActiveSheet.Visible = False
End Sub

Sub ViewChart()
    Sheets("ResultsChart").Visible = True
    Worksheets("Model").Visible = False
End Sub

Sub ReturnSheet()
    Worksheets("Model").Visible = True
    Sheets("ResultsChart").Visible = False
End Sub

```

Figure CS20.17 The navigational procedures.

 <p>Summary</p>	Main	Initializes the application and prompts the user with the historical data form.
	Demo	Copies the demo year and data to the historical data sheet.
	ClearPrev	Clears the previous values from all the sheets and initializes the special factors arrays.
	Historical data form procedures	Determine how the user will enter the historical data.
	Year spin button event procedure	Updates the calendar dates, the month, and the weekday values for the selected year.
	Continue	Prepares the model on the solution sheet and shows the user the special factors form.
	Special factors form procedures	Record the special factors selected by the user.
	RecordSF	Marks the special factors on the calendar dates and in the model.
	RunSolver	Prepares the model with the formulas, runs the Solver, highlights the outliers, and updates the graph data.
	Navigational procedures	For the "End," "View Chart," and "Return to Results" buttons.

CS20.5 *Re-solve Options*

The user can re-solve this application by pressing the "Re-solve" button on the solution sheet. This button is assigned to the "Re-solve" procedure, which clears the previous factor values and re-shows the special factors form. (See Figure CS20.18.) This option allows the user to modify his or her selection of special factors based on the previous results. For example, the user may notice that outliers can be found on a periodic basis and therefore want to add a new periodic special factor to account for this pattern in the data. Once the user updates the special factors, they are recorded in the model and the "RecordSF" and "RunSolver" procedures are re-called. The user stays on the solution sheet to view the updated results.

```

Sub Resolve() 'resolve the problem with updated special factors
    Range("SPFacStart").Offset(-2, 2).ClearContents
    Range("SPFacStart").Offset(-2, 4).ClearContents
    Range("SPFacStart").Offset(-2, 6).ClearContents
    frmSpecFact.Show

    Application.ScreenUpdating = False
    Range("SPValues").Columns.Hidden = False
    Range(Range("SFStart").Offset(1, 0), Range("SFStart").Offset(1, 6).End(xlDown)).ClearContents
    Range(Range("DataStart").Offset(1, 3), Range("DataStart").Offset(1, 3).End(xlDown)).ClearContents
    Range("CFactor").ClearContents
    Range("MonthFactors").ClearContents
    Range("WeekFactors").ClearContents
    Range("SpecialFactors").ClearContents
    Range("MSE").Interior.ColorIndex = 2

    Call RecordSF
    Call RunSolver

    Range("A1").Select
    Application.ScreenUpdating = True
End Sub

```

Figure CS20.18 The “Re-solve” procedure.

 Summary	<p>“Re-solve” button on the solution sheet</p> <p>Re-shows the special factors form and re-optimizes the factor values and the customer forecast.</p>
---	--

CS20.6 *Summary*

- This application is to forecast stochastic customer arrival while considering a set of factors that influence the forecast.
- This application requires four worksheets: the welcome sheet, the historical data sheet, the solutions sheet, and the solutions chart sheet.
- For this application’s user interface, we use navigational buttons, controls on the worksheet, and two user forms.
- Several procedures for this application collect the historical data and the special factors for the model and solve the model to forecast the customer arrivals.
- The user can re-solve this application by pressing the “Re-solve” button on the solution sheet. This selection re-shows the special factors form and re-optimizes the factor values and the customer forecast.

CS20.7 *Extensions*

- Add a dynamic element to the special factors form by selecting the first value in the combo boxes for any corresponding special factor check box.
- Ensure that the user’s previous selections on the special factors form appear when the form is re-displayed during the re-solve option. In other words, if the

user had previously selected a date or entered a period special factor, then it should already be checked and entered when the form appears again.

- Attempt to solve this problem using multiple regression instead of a non-linear programming model.